
Resumo

Mineração de Dados Relacional (MDR) é um processo de descoberta de

novos padrões e conhecimento útil que envolve múltiplas tabelas (relações)

de uma base de dados relacional. A maioria das abordagens de MDR

é baseada em Programação Lógica Indutiva (PLI). PLI é definida como

a intersecção de Aprendizado de Máquina indutivo e programação lógica

e tem se expandido para cobrir todas as tarefas de MDR. O sistema de

PLI Aleph, foco de estudo deste trabalho, possui, além das caracteŕısticas

comuns aos sistemas de PLI, a capacidade de emular funcionalidades de

vários outros sistemas de PLI, tais como, regras de classificação relacionais,

árvores de decisão relacionais, regras de associação relacionais, entre outras.

Neste trabalho são descritas maneiras de como emular diversos sistemas de

PLI utilizando Aleph, descrevendo o ajuste de parâmetros necessário e a

escolha dos diversos comandos dispońıveis.

Palavras-Chave: Aprendizado de Máquina, Programação Lógica Indutiva,

Aleph.

3 Uso Avançado do Aleph 24

3.1 Comparação entre a Construção de Teorias induce/0 e induce cover/0 25

1 Introdução

Mineração de Dados (MD) é um processo cujo principal objetivo é extrair padrões de

bases de dados. As principais etapas do processo de Mineração de Dados (MD) são:

o pré-processamento de dados, na qual procura-se melhorar a qualidade dos dados; a

etapa posterior de extração de padrões, na qual são freqüentemente utilizados algo-

ritmos de aprendizado e a etapa de pós-processamento desses padrões (conhecimento)

na qual o objetivo é selecionar ou melhorar os padrões encontrados. A maioria das

abordagens de MD assume que os dados estão em uma única tabela. Esse tipo de

dados pode ser manipulado pelos algoritmos de Aprendizado de Máquina (AM) pro-

posicional que utilizam uma linguagem de representação proposicional para descrever

exemplos, tais como, uma tabela atributo-valor. Por outro lado, as abordagens de Mi-

neração de Dados Relacional (MDR), muitas das quais são baseadas em Programação

Lógica Indutiva (PLI), realizam a busca por padrões que envolvem múltiplas tabelas

(relações) de uma base de dados relacional. Esses métodos de Mineração de Dados

Relacional (MDR) utilizam uma linguagem de representação relacional baseada em

lógica de primeira ordem (Džeroski, 2003).

Assim como muitos dos algoritmos utilizados na etapa de extração de padrões no pro-

cesso de MD vêm da área de Aprendizado de Máquina (AM), muitos dos algoritmos de

MDR vêm da área de Programação Lógica Indutiva (PLI), que consiste da intersecção

do AM indutivo e da programação lógica. O objetivo da PLI é aprender programas

lógicos a partir de exemplos e conhecimento do domı́nio (Lavrač & Džeroski, 2001).

Nos últimos anos, a PLI tem se expandido para cobrir todas as tarefas de MDR.

Os tipos mais comuns de técnicas utilizadas em AM proposicional para descrever os

padrões encontrados têm sido estendidos para suas versões relacionais tais como, re-

gras de classificação relacionais, árvores de decisão relacionais, regras de associação

relacionais, entre outros, descritos brevemente a seguir.

A tarefa de indução de regras relacionais é a tarefa mais freqüentemente empregada em

PLI. Essa é a tarefa do aprendizado de definições lógicas de relações na qual tuplas que

pertencem ou não pertencem à relação alvo são dadas como exemplos de treinamento.

A partir dos exemplos de treinamento, PLI induz um programa lógico (definição do

predicado) que define a relação alvo em termos de outras relações que são dadas como

conhecimento de fundo. Alguns exemplos de sistemas de PLI que realizam indução de

regras são o Progol (Muggleton, 1995) e o FOIL (Quinlan, 1990).

Nas árvores de decisão proposicionais, nós internos são testes e os nós folha são as

predições para os valores da classe. As árvores de decisão relacionais possuem uma

estrutura muito parecida com as proposicionais. A maior diferença para as árvores

relacionais está nos testes que aparecem nos nós internos. No caso relacional, os testes

são consultas lógicas. O sistema de PLI Tilde (Blockeel, 1998) faz indução de árvores

de decisão relacionais.

As regras de associação caracterizam o quanto a presença de um conjunto de itens

de uma base de dados implica a presença de algum outro conjunto distinto de itens

nos mesmos registros. No caso das bases de dados relacionais, regras de associação

relacionais poderão ser descobertas. O sistema WARMR (Dehaspe & De Raedt, 1997)

realiza indução de regras de associação relacionais.

O sistema Aleph, A Learning Engine for Proposing Hypotheses (Srinivasan, 2000),

é um sistema de PLI escrito em Prolog, o qual foi inicialmente conhecido como P-

Progol. O sistema Aleph, foco deste trabalho, foi desenvolvido com o intuito de ser

um protótipo para explorar idéias de PLI. O Aleph possui uma poderosa linguagem de

representação que permite representar expressões complexas e incorporar novo conhe-

cimento do domı́nio com facilidade. Além dessas caracteŕısticas comuns aos sistemas

de PLI, o Aleph possui outras caracteŕısticas muito interessantes, tais como: permitir

escolher qual a ordem de geração das regras (geral para o espećıfico ou espećıfico para o

geral); mudar a função de avaliação (entropia, precisão, Laplace entre outras) e a ordem

de busca (Hill-Climbing, Branch-and-Bound, entre outras). Utilizando os parâmetros

dispońıveis no Aleph para modificação de funções de busca, funções de avaliação,

além de diversos parâmetros espećıficos para determinadas tarefas, é posśıvel emular

funcionalidades de vários outros sistemas de PLI, tais como: Progol (Muggleton, 1995),

FOIL (Quinlan, 1990), FORS (Karalič & Bratko, 1997), Indlog (Camacho, 1994),

MIDOS (Wrobel, 1997), Tilde (Blockeel, 1998) e WARMR (Dehaspe & De Raedt,

1997). Devido a essa versatilidade do Aleph, foi decidido utilizá-lo em nossas pesquisas

relacionadas a aprendizado relacional.

Porém, por falta de documentação adequada sobre o Aleph, a maneira de como emular

esses outros sistemas não está o suficientemente clara. É necessário saber como ajustar

adequadamente os parâmetros, escolher corretamente os diversos comandos dispońıveis

para se construir uma teoria e, até mesmo, saber quais dados podem ser utilizados para

cada tarefa. Todas essas dúvidas na utilização do Aleph motivaram a realização deste

trabalho, que tem como objetivo entender a ação dos comandos e parâmentros mais

importantes do Aleph, bem como documentar o funcionamento dos mesmos.

Este trabalho está organizado da seguinte forma: na Seção 2 são apresentados alguns

conceitos sobre PLI, caracteŕısticas básicas e de funcionamento do sistema de PLI

Aleph, além de um exemplo de execução utilizando Aleph; na Seção 3 são apresentadas

as formas dispońıveis para se construir uma teoria, alterar funções de busca e avaliação,

e como emular as funcionalidades do Aleph para expressar os padrões utilizando como

linuagens de descrição árvores de decisão e regras de associação relacionais, além dos

métodos de busca estocásticos. Descreve-se ainda como o usuário pode interferir no

processo de aprendizado; por fim, na Seção 4 são apresentadas algumas considerações

finais sobre este trabalho.

2 Programação Lógica Indutiva e Aleph

PLI é uma área de pesquisa em AM definida como a intersecção de AM indutivo e

programação lógica (Lavrač & Flach, 2001). Em programação lógica, o conceito a

ser aprendido é um programa lógico, i.é., um conjunto de cláusulas de Horn. Da

lógica computacional, PLI herda seu formalismo representacional, várias técnicas bem

estabelecidas e uma profunda base teórica. Do AM indutivo, herda uma abordagem

experimental e orientação para aplicações práticas, tais como o desenvolvimento de

ferramentas e técnicas para induzir hipóteses a partir de exemplos e sintetizar novos

conhecimentos a partir de experiências (Muggleton & Raedt, 1994).

A habilidade dos seres humanos de fazer generalizações a partir de alguns fatos, ou

descobrir padrões em coleções de observações aparentemente caóticas, é atingida por

meio do aprendizado por indução. A indução é a forma de inferência lógica que per-

mite obter conclusões gerais a partir de um conjunto particular de exemplos, ou casos

observados. É caracterizada como o racioćınio que parte do espećıfico para o geral,

da parte para o todo. No aprendizado por indução, o aprendiz adquire um conceito

fazendo inferências indutivas sobre os fatos apresentados. A indução pode ser vista

como o inverso da dedução. Hipóteses geradas pela inferência indutiva podem ou não

preservar a verdade, ao contrário da dedução na qual, se todas as premissas forem

verdadeiras, a conclusão é verdadeira. Há uma segunda diferença entre as hipóteses

dedutivas e indutivas. Na dedução, toda a informação do conteúdo factual da conclu-

são já está, pelo menos implicitamente, nas premissas. Em contraste, na indução a

conclusão contém informação que excede a das premissas, como mostrado a seguir.

Exemplo de hipótese dedutiva

Todo mamı́fero tem coração.

Todos os felinos são mamı́feros.

−→ Todos os felinos têm coração.

Exemplo de hipótese indutiva

Todos os felinos que foram observados tinham coração.

−→ Todos os felinos têm coração.

PLI pode ser vista como o estudo de métodos de aprendizado para dados e regras

que são representados como predicados lógicos de primeira ordem. Predicados lógicos

permitem variáveis quantificadas e relações, e podem representar conceitos que não

podem ser expressos usando uma linguagem de descrição proposicional. Uma base de

dados relacional pode ser facilmente traduzida em lógica de primeira ordem e ser usada

como fonte de dados para PLI.

Como exemplo, considere as seguintes regras escritas na sintaxe Prolog, as quais defi-

nem a relação tio/2.

tio(X,Y):- irmao(X,Z), progenitor(Z,Y).

tio(X,Y):- marido(X,Z),irma(Z,W), progenitor(W,Y).

O objetivo da PLI é inferir regras, como as apresentadas acima, dada uma base de

dados com fatos e definições lógicas de outras relações. Por exemplo, um sistema

de PLI pode aprender as regras para tio/2, chamado de predicado alvo, dado um

conjunto de exemplos positivos e negativos das relações tio/2 e um conjunto de fatos

para as relações progenitor/2, irmao/2, irma/2 e marido/2, chamadas de predicados

de conhecimento de fundo para os membros de uma famı́lia (Mooney et al., 2002).

Em termos um pouco mais formais, dado um programa lógico B que modela o conhe-

cimento de fundo e um conjunto de termos ground G que representa os exemplos de

treinamento, PLI constrói um programa lógico H, tal que B ∧H |= G. Se o operador

dedutivo usado for a resolução, então o operador indutivo necessário para resolver a

equação é o operador de resolução inversa (Muggleton, 1995).

Atualmente existem diversos sistemas de PLI, entre eles o sistema Aleph (Srinivasan,

2000). Esse sistema de PLI possui uma poderosa linguagem de representação além de

diversos recursos que lhe permite simular funcionalidades de vários outros sistemas de

PLI existentes, como mencionado anteriormente. O sistema Aleph é o foco principal

deste trabalho e será apresentado com maiores detalhes nas próximas seções.

2.1 Caracteŕısticas Básicas do Sistema Aleph

Nesta seção são explicados em maiores detalhes várias caracteŕısticas e comandos uti-

lizados para a execução do sistema Aleph.

2.1.1 Como Iniciar o Sistema

Aleph é um sistema escrito em Prolog que deve ser utilizado com o compilador Yap

Prolog. Ainda que o compilador SWI Prolog poderia ser utilizado, experimentos reali-

zados mostraram que sua utilização prejudica algumas das funcionalidades do sistema.

Para carregar o Aleph, deve-se inicialmente carregar o sistema Yap e, em seguida,

chamá-lo através do comando [aleph]., como mostra o exemplo a seguir:

fapaiva@penelope:~$ yap

% Restoring file /usr/local/lib/Yap/startup

YAP version Yap-5.1.1

?- [aleph].

% consulting /home/fapaiva/aleph.pl...

A L E P H

Version 5

Last modified: Sun Jun 4 10:51:31 UTC 2006

Manual: http://www.comlab.ox.ac.uk/oucl/groups/machlearn/Aleph/index.html

% consulted /home/fapaiva/aleph.pl in module user, 140 msec 1138688 bytes

yes

?-

2.1.2 Sobre o Sistema

Aleph é um sistema de PLI que constrói incrementalmente as cláusulas de uma hipó-

tese, iterativamente, por meio do seguinte algoritmo básico:

1. Seleciona um exemplo positivo do conjunto de exemplos de treinamento para ser

generalizado. Se não existirem mais exemplos, pára;

2. Saturação: constrói a cláusula mais espećıfica, não ground, que implique o exem-

plo selecionado no passo anterior. Isso é feito por repetidas aplicações do ope-

rador de resolução inversa sobre o exemplo, até que todos seus termos ground

tenham sido substitúıdos por variáveis, as quais são distribúıdas pelo corpo da

cláusula, até que o exemplo positivo ground original, e somente ele, seja co-

berto pela cláusula bottom. A cláusula minimamente generalizada é chamada de

cláusula bottom;

3. Redução: realiza uma busca por uma “boa” cláusula entre a cláusula maxima-

mente geral (maximally general), a cláusula vazia e a cláusula bottom maxima-

mente espećıfica (maximally specific). O espaço entre a cláusula vazia e a cláusula

bottom é parcialmente ordenado pela subsunção-θ e a busca prossegue pelo reti-

culado definido por essa ordenação parcial. A “boa qualidade” de cada cláusula

encontrada ao longo do caminho de busca é medida utilizando uma função de

avaliação;

4. Cover Removal : adiciona a cláusula de melhor qualidade, chamada de best clause,

à teoria e remove todos os exemplos positivos cobertos por ela da base de exem-

plos;

5. Repete até que todos os exemplos positivos sejam cobertos.

O Aleph utiliza a informação contida nos seguintes três arquivos para construir uma

teoria:

• file.b: contém a informação sobre o conhecimento do domı́nio (intensional e

extensional), as restrições de busca e linguagem (declarações de modo, determi-

nations), restrições de tipos e parâmetros;

• file.f: contém os exemplos positivos (somente fatos ground);

• file.n: contém os exemplos negativos (somente fatos ground). Pode não existir

esse arquivo, já que Aleph pode realizar aprendizado utilizando somente exemplos

positivos.

2.1.3 Declarações de Modo

As declarações de modo no arquivo file.b descrevem as relações (predicados) entre

objetos de dados e tipos desses dados. Essas declarações permitem também informar

ao Aleph se a relação pode ser utilizada na cabeça (declarações modeh) ou no corpo

(declarações modeb) das regras geradas. Na declaração de modo também são descritos

o tipo dos argumentos e o número de instanciações de cada predicado. As declarações

têm o seguinte formato:

mode(Numero_Chamadas,Modo).

O Numero_Chamadas, ou como é mais conhecido, o recall, determina o limite do número

de instanciações alternativas de um predicado. Uma instanciação do predicado é uma

substituição de tipos declarados no arquivo .b para cada variável ou constante. O

recall pode ser qualquer número positivo n ≥ 1 ou ’*’.

Se é conhecido que há somente um certo número de instanciações posśıveis para um

determinado predicado, é posśıvel informar isso pelo recall. Por exemplo, para uma de-

claração do predicado progenitor_de(Pai,Filho) poderia ser dado um recall igual a

2, pois todas as pessoas terão no máximo dois progenitores, enquanto que o recall seria

4 para uma declaração do predicado avo(Avo,Neto). O recall ’*’ é utilizado quando

não há limite para o número de instanciações de um predicado. Por exemplo, em

prinćıpio, não há limite para o número de ancestrais que uma pessoa pode ter.

A segunda parte da declaração do mode/2, denominada de Modo, indica o formato do

predicado que será utilizado e tem o seguinte formato:

predicado(Tipo_Argumento1,Tipo_Argumento2,...,Tipo_Argumenton).

Para o aprendizado da relação sogra_de(Sogra,Genro) com conhecimento do domı́nio

descrito, por exemplo, pelas relações progenitor_de(Mae,Filha) e esposa_de(Esposa,Marido)

as declarações de modo podem ser:

:- modeh(1,sogra_de(+mulher,+homem)).

:- modeb(*,progenitor_de(+mulher,-mulher)).

:- modeb(1,esposa_de(+mulher,+homem)).

A declaração modeh/2 indica o predicado que irá compor a cabeça das regras. O va-

lor de RecallNumber é 1 para esse predicado, pois o predicado pode ter uma única

resposta (sim ou não), dados os dois argumentos de entrada. No exemplo, modeh/2

informa que a cabeça das regras deve ser sogra_de(Sogra,Genro), onde Sogra é

do tipo mulher e Genro é do tipo homem. As declarações modeb/2 indicam que

as regras geradas podem ter no corpo os predicados progenitor_de(Mae,Filha) e

esposa_de(Esposa,Marido), nos quais Mae, Filha e Esposa são do tipo mulher e

Marido é do tipo homem. O valor de RecallNumber para o predicado progenitor_de(Mae,Filha)

é igual a ’*’, pois não se sabe quantas filhas uma mãe pode ter. Para o predicado

esposa_de(Esposa,Marido) esse parâmetro tem valor ’1’, pois apenas uma das filhas

será a esposa.

Os tipos dos argumentos dos predicados podem ser estruturados ou simples. Os tipos

estruturados são da forma f(..), no qual f é um śımbolo de função. Um exemplo de

argumento estruturado é:

:- modeh(1,member(+number,[+number|+lista])).

Os tipos simples de argumentos podem ser: +, - ou #. O śımbolo ’+’ indica que o

argumento do predicado é uma variável de entrada. O śımbolo ’-’ indica uma variável

de sáıda e o śımbolo # indica que esse argumento é uma constante. A utilização desses

tipos de variáveis devem seguir as regras listadas a seguir:

• Variável de Entrada (+): Qualquer variável de entrada do tipo T em um literal

do corpo Bi ( modeb ) deve aparecer ou como uma variável de sáıda do tipo T

em um literal do corpo antes de Bi, ou como uma variável de entrada do tipo T

na cabeça, como mostra o exemplo a seguir:

:- modeh(1, sogra_de(+mulher,+homem)).

:- modeb(*, progenitor_de(+mulher,-mulher)).

:- modeb(1, esposa_de(+mulher,+homem)).

A variável de entrada do tipo homem do predicado esposa_de/2 vem da variável

de mesmo tipo da cabeça. A variável de entrada do tipo mulher do predi-

cado esposa_de/2 vem ou da variável de sáıda de mesmo tipo do predicado

progenitor_de/2 ou da variável de entrada de mesmo tipo da cabeça.

• Variável de Sáıda (-): Qualquer variável de sáıda do tipo T na cabeça (modeh)

deve aparecer como uma variável de sáıda do tipo T em um literal do corpo Bi.

Qualquer literal que possua variável de sáıda deve ter, pelo menos, uma variável

de entrada. Como mostra o exemplo a serguir:

:- modeh(1, sogra_de(+mulher,-homem)).

:- modeb(1, progenitor_de(+mulher,-mulher)).

:- modeb(1, esposa_de(+mulher,-homem)).

A variável de sáıda do tipo homem da cabeça vem da variável de sáıda de mesmo

tipo do predicado esposa_de/2.

• Constante: Qualquer argumento denotado por #T só pode ser substitúıdo por

átomos ground.

2.1.4 Tipos

Os tipos devem ser especificados para cada argumento dos predicados utilizados na

construção de uma hipótese. Para o Aleph, os tipos são só nomes e a declaração de

tipos nada mais é que um conjunto de fatos. Aleph reconhece argumentos de tipos

diferentes e os trata distintamente. Por exemplo, a descrição de objetos dos tipos

homem e mulher poderia ser

mulher(jane).

mulher(sueli).

homem(henrique).

homem(junior).

...

2.1.5 Declaração dos Determinations

O Aleph utiliza determination/2 para declarar os predicados que podem ser usados
para construir uma hipótese. Essa declaração tem o formato:

determination(Pred_Alvo/Aridadea,Pred_Corpo/Aridadec).

O primeiro argumento é o nome do predicado alvo e sua aridade, isto é, o predicado

que irá aparecer na cabeça da regra induzida. O segundo argumento consiste do nome

e aridade de um predicado que pode aparecer no corpo da cláusula. Por exemplo, para

o aprendizado da relação sogra_de(Sogra,Genro), uma declaração seria:

:-determination(sogra_de/2,progenitor_de/2).

:-determination(sogra_de/2,esposa_de/2).

Para que um predicado seja utilizado na construção de uma hipótese, é necessário que

ele tenha declarações tanto de modeb/2 quanto de determination/2. Caso uma das

declarações esteja ausente, o predicado não será utilizado na construção da hipótese.

2.1.6 Exemplos Positivos e Negativos

Exemplos positivos e negativos do conceito a ser aprendido devem ser apresentados
na forma extensional (fatos) nos arquivos com extensão .f e .n respectivamente. Por
exemplo, para o aprendizado do conceito sogra_de(Sogra,Genro) podeŕıamos ter a
declaração dos seguintes exemplos positivos no arquivo com extensão .f:

sogra_de(jane,henrique).

sogra_de(sueli,joao).

E os seguintes exemplos negativos no arquivo com extensão .n:

sogra_de(jane,joao).

sogra_de(sueli,henrique).

Vale observar que Aleph pode realizar aprendizado somente com exemplos positivos,

nesse caso o arquivo .n não é utilizado.

2.1.7 Construção da Cláusula Bottom

Como já mencionado anteriormente, a cláusula bottom, constrúıda no passo de satura-

ção, é a cláusula mais espećıfica que implique o exemplo selecionado. Ela é importante,

pois é a partir dela que o Aleph constrói as posśıveis hipóteses para expressar o conceito

a ser aprendido. Sua corretude depende do parâmetro recall e das relações entre os

tipos dos argumentos dos predicados presentes nas declarações de modo. Para ilustrar

a evolução do passo de saturação, bem como a finalidade dos parâmetros utilizados

nesse passo, considere o seguinte conteúdo do arquivo .b

:- modeh(1, sogra_de(+mulher,-homem)).

:- modeb(*, progenitor_de(+mulher,-mulher)).

:- modeb(1, esposa_de(+mulher,-homem)).

:- determination(sogra_de/2,progenitor_de/2).

:- determination(sogra_de/2,esposa_de/2).

%declaraç~ao dos tipos

homem(pai1).

homem(marido1).

homem(marido2).

mulher(mae1).

mulher(filha11).

mulher(filha12).

mulher(neta11).

%conhecimento de fundo

progenitor_de(mae1, filha11).

progenitor_de(pai1, filha11).

progenitor_de(mae1, filha12).

progenitor_de(pai1, filha12).

progenitor_de(filha11, neta11).

esposa_de(mae1, pai1).

esposa_de(filha11, marido1).

esposa_de(filha12, marido2).

o seguinte conteúdo do arquivo de exemplos positivos (.f)

sogra_de(mae1, marido1).

sogra_de(mae1, marido2).

e o seguinte conteúdo do arquivo de exemplos negativos (.n).

sogra_de(mae1, pai1).

sogra_de(filha11, marido2).

sogra_de(filha11, marido1).

A saturação do exemplo positivo sogra_de(mae1, marido1) segue os seguintes passos:

1. Substituição dos argumentos de entrada da cabeça pelos valores correspondentes

do exemplo positivo a ser saturado. Os argumentos de sáıda da cabeça são

substitúıdos por variáveis. As constantes são sempre substitúıdas por átomos

ground. No exemplo sogra_de(mae1, marido1) a variável do tipo mulher é

substitúıda por mae1 e a variável de sáıda do tipo homem é substitúıda por uma

variável de nome A. O resultado da substituição é: sogra_de(mae1, A).

2. São escolhidos os predicados do corpo que tenham argumentos de entrada do

mesmo tipo que os argumentos de entrada da cabeça. Os argumentos de entrada

desses predicados são substitúıdos pelos mesmos valores cujos argumentos de

entrada da cabeça foram substitúıdos no passo acima. Os argumentos de sáıda

desses predicados são substitúıdos por variáveis — Figura 1. Então, procura-se

por fatos no conhecimento do domı́nio que unifiquem com esses predicados. Essa

busca por fatos pode ocorrer de uma a N vezes, dependendo do valor do parâme-

tro recall, para cada argumento de entrada. Todos os fatos no conhecimento do

domı́nio que unificam com o predicado informam quais valores as variáveis desse

predicado podem assumir. Por exemplo, a variável B do tipo mulher do predi-

cado progenitor_de/2 pode ser substiúıda pelos valores filha11 e filha12. E

a variável C do tipo homem do predicado esposa_de/2 pode ser substitúıda pelo

valor pai1 — Figura 2.

Figura 1: Substituições dos argumentos dos predicados a partir da unificação do exem-
plo com a cabeça.

3. Os valores encontrados para as variáveis de sáıda são utilizados para a busca por

fatos de predicados do corpo que possuem variáveis de entrada do mesmo tipo.

Por exemplo:

Figura 2: Substituições dos argumentos dos predicados a partir das unificações com
fatos do conhecimento do domı́nio.

esposa_de(filha11, C). −→ esposa_de(filha11, marido1).

esposa_de(filha12, C). −→ esposa_de(filha12, marido2).

Nesse caso, a variável C do tipo homem só pode assumir o valor marido1, pois

a substituição pelo valor marido2 não implica o exemplo sogra_de(mae1, ma-

rido1). Esse é um caso da regra da variável de sáıda em que uma variável de

sáıda da cabeça deve vir de uma variável de sáıda do mesmo tipo do corpo.

4. Ao final, são adicionados à cláusula bottom os predicados encontrados cujos valo-

res combinam com os valores do exemplo saturado (condição para que o exemplo

seja coberto) e todos os valores encontrados para os argumentos dos predicados

são substitúıdos por variáveis. Os argumentos substitúıdos nos passos anteriores

por valores idênticos, são agora substitúıdos por variáveis de mesmo nome. Por

exemplo:

sogra_de(A,B):-

progenitor_de(A,C),progenitor_de(A,D),

progenitor_de(A,E),esposa(C,B).

De acordo com as regras listadas na Seção 2.1.3 na página 8, para cada tipo de variável

(entrada, sáıda ou constante), a relação entre os tipos de variáveis dos argumentos de

cada predicado se dá como apresentado na Figura 3.

O sistema Aleph procura pelos fatos no conhecimento do domı́nio na ordem em que

aparecem no arquivo .b. Nesse caso, se o valor do parâmetro recall fosse 1 para o pre-

dicado progenitor_de/2, o resultado da cláusula bottom ainda estaria correto, pois

o primeiro fato encontrado no conhecimento do domı́nio seria o da filha11 que é

Figura 3: Relação entre os tipos de variáveis dos argumentos de cada predicado.

casada com o marido1. Porém, trocando os fatos progenitor_de(mae1, filha11)

e progenitor_de(mae1, filha12) de posição no arquivo .b (veja abaixo a altera-

ção no arquivo .b), o primeiro fato encontrado no conhecimento do domı́nio seria

progenitor_de(mae1, filha12), que não é casada com marido1. Não encontrando

a filha correta, o sistema não encontraria um fato para esposa_de/2 que casasse com

o exemplo, e assim, a cláusula bottom gerada seria: sogra_de(A,B). Portanto, a esco-

lha do valor para o parâmetro recall pode ser a causa de uma cláusula bottom gerada

incorretamente.

:- modeh(1, sogra_de(+mulher,-homem)).

:- modeb(1, progenitor_de(+mulher,-mulher)).

:- modeb(1, esposa_de(+mulher,-homem)).

:- determination(sogra_de/2,progenitor_de/2).

:- determination(sogra_de/2,esposa_de/2).

%declaraç~ao dos tipos

homem(pai1).

homem(marido1).

homem(marido2).

mulher(mae1).

mulher(filha11).

mulher(filha12).

mulher(neta11).

%conhecimento de fundo

%progenitor_de(mae1, filha12) aparece antes de progenitor_de(mae1, filha11).

%Isso implica que progenitor_de(mae1, filha12) será o primeiro e único fato

%(único, pois Recall é 1) encontrado para o predicado progenitor_de/2.

progenitor_de(mae1, filha12).

progenitor_de(pai1, filha11).

progenitor_de(mae1, filha11).

progenitor_de(pai1, filha12).

progenitor_de(filha11, neta11).

esposa_de(mae1, pai1).

esposa_de(filha11, marido1).

esposa_de(filha12, marido2).

2.1.8 Parâmetros

O Aleph permite ao usuário determinar uma variedade de restrições no espaço das
posśıveis hipóteses a serem aprendidas, bem como na busca realizada nesse espaço
de posśıveis hipóteses, pela definição de novos valores dos parâmetros dispońıveis. O
predicado set/2 permite ao usuário definir o valor do parâmetro Parametro

set(Parametro,Valor).

O valor corrente de um determinado parâmetro pode ser obtido usando

setting(Parametro,Valor).

enquanto que

noset(Parametro).

retorna o valor desse parâmetro para seu valor padrão. O uso de diversos parâmetros

no Aleph será exemplificado ao longo deste trabalho.

2.2 Exemplo de Execução

Nesta seção é apresentado um exemplo de como utilizar o Aleph, ou seja, qual a

seqüência básica de comandos utilizados para se construir uma teoria e como o Aleph

irá responder a esses comandos. Para ilustrar, será utilizado o exemplo dos trens de

Michalski (Michalski et al., 1989), no qual, dada uma seqüência de trens — Figura 4 —

cada trem tem um número de vagões e cada vagão pode ter diferentes propriedades,

indicadas pelo desenho de uma figura com um determinado formato no seu interior.

Também sabe-se que o trem está viajando para o leste (East) ou para o oeste (West).

O problema é encontrar uma regra capaz de predizer, dadas as propriedades dos seus

vagões, se o trem está viajando para leste; esse trem é chamado de eastbound.

A Tabela 1 mostra a declaração dos exemplos positivos e negativos nos arquivos cor-

respondentes trens.f e trens.n. O próximo passo é definir o arquivo trens.b com

16



Figura 4: Trens que viajam para Leste e Oeste (Michalski et al., 1989)

o conhecimento de fundo e as diretivas para a execução do Aleph. A primeira diretiva

set/2 indica que as regras que constituem a hipótese devem cobrir, no mı́nimo, dois

exemplos positivos. Exceto essa definição de minpos, cujo valor padrão é 1, Aleph

foi executado com os parâmetros padrão. O conceito a ser aprendido é denominado

eastbound/1, portanto deve ser declarado com modeh/2, e os predicados que podem

aparecer na regra devem ser declarados com modeb/2, utilizando os tipos de dados

dos argumentos desses predicados, definidos mais a frente, bem como as instanciações

desses argumentos, como mostrado a seguir.

Ex.(+) trens.f Ex.(-) trens.n
1 eastbound(east1). 6 eastbound(west6).
2 eastbound(east2). 7 eastbound(west7).
3 eastbound(east3). 8 eastbound(west8).
4 eastbound(east4). 9 eastbound(west9).
5 eastbound(east5). 10 eastbound(west10).

Tabela 1: Arquivos trens.f e trens.n

:- set(minpos,2).

:- modeh(1,eastbound(+train)).

:- modeb(1,short(+car)).

:- modeb(1,closed(+car)).

:- modeb(1,long(+car)).

:- modeb(1,open_car(+car)).

:- modeb(1,double(+car)).

:- modeb(1,jagged(+car)).

:- modeb(1,shape(+car,#shape)).

:- modeb(1,load(+car,#shape,#int)).

:- modeb(1,wheels(+car,#int)).

:- modeb(*,has_car(+train,-car)).

Também devem ser declarados os determination/2, especificando os predicados que

podem aparecer na cabeça e no corpo das regras com essa cabeça, como mostrado a

seguir.

:- determination(eastbound/1,short/1).

:- determination(eastbound/1,closed/1).

:- determination(eastbound/1,long/1).

:- determination(eastbound/1,open_car/1).

:- determination(eastbound/1,double/1).

:- determination(eastbound/1,jagged/1).

:- determination(eastbound/1,shape/2).

:- determination(eastbound/1,wheels/2).

:- determination(eastbound/1,has_car/2).

:- determination(eastbound/1,load/3).

Para definir o tipo de dados car/1 enumerando os posśıveis valores que pode assumir,

esses valores foram definidos considerando primeiro o trem e segundo o vagão. Por

exemplo car(car_14) refere-se ao valor do tipo car para o trem 1 vagão 4. Tam-

bém são usados shape/1 que define os posśıveis formatos das figuras relacionadas às

propriedades dos vagões, bem como train/1 que define os trens.

car(car_11). car(car_12). car(car_13). car(car_14).

car(car_21). car(car_22). car(car_23).

car(car_31). car(car_32). car(car_33).

car(car_41). car(car_42). car(car_43). car(car_44).

car(car_51). car(car_52). car(car_53).

car(car_61). car(car_62).

car(car_71). car(car_72). car(car_73).

car(car_81). car(car_82).

car(car_91). car(car_92). car(car_93). car(car_94).

car(car_101). car(car_102).

shape(elipse). shape(hexagon). shape(rectangle). shape(u_shaped).

shape(triangle). shape(circle). shape(nil).

train(east1). train(east2). train(east3). train(east4). train(east5).

train(west6). train(west7). train(west8). train(west9). train(west10).

O próximo passo consiste em definir cada um dos 10 trens. Por exemplo, o trem 1 é

definido como

short(car_12). % vag~ao 2 é curto

closed(car_12). % vag~ao 2 é fechado

long(car_11). % vag~ao 1 é longo

long(car_13). % vag~ao 3 é longo

short(car_14). % vag~ao 4 é longo

open_car(car_11). % vag~ao 1 é aberto

open_car(car_13). % vag~ao 3 é aberto

open_car(car_14). % vag~ao 4 é aberto

shape(car_11,rectangle). % vag~ao 1 tem forma rectangular

shape(car_12,rectangle). % vag~ao 2 tem forma rectangular

shape(car_13,rectangle). % vag~ao 3 tem forma rectangular

shape(car_14,rectangle). % vag~ao 4 tem forma rectangular

load(car_11,rectangle,3). % vag~ao 1 carrega 3 retângulos

load(car_12,triangle,1). % vag~ao 1 carrega 1 triângulo

load(car_13,hexagon,1). % vag~ao 1 carrega 1 hexágono

load(car_14,circle,1). % vag~ao 1 carrega 1 cı́rculo

wheels(car_11,2). % vag~ao 1 tem 2 rodas

wheels(car_12,2). % vag~ao 1 tem 2 rodas

wheels(car_13,3). % vag~ao 1 tem 3 rodas

wheels(car_14,2). % vag~ao 1 tem 2 rodas

has_car(east1,car_11). % o vag~ao 1 é do trem east1

has_car(east1,car_12). % o vag~ao 2 é do trem east1

has_car(east1,car_13). % o vag~ao 3 é do trem east1

has_car(east1,car_14). % o vag~ao 4 é do trem east1

Os outros 9 (nove) trens devem ser descritos de maneira semelhante no arquivo trens.b.

Considerando que todos os trens estão descritos nesse arquivo, os três arquivos são lidos

executando read_all/1 como mostrado a seguir.

?- read_all(train).

Com esse comando também é posśıvel informar o caminho onde Aleph deve consultar
os arquivos. Por exemplo:

?- read_all(’/home/mariza/examples-aleph/trains/train’).

Após consultar os arquivos, Aleph está pronto para induzir uma teoria. Para que o

Aleph execute seu algoritmo básico, utiliza-se o comando induce/0:

?- induce.

Seguindo os passos do algoritmo básico de funcionamento o primeiro passo é a seleção
de um exemplo para ser generalizado, gerando a seguinte execução:

[select example] [1]

[sat] [1]

[eastbound(east1)]

[select example] indica o número do exemplo selecionado para ser generalizado.

[sat] indica o número do exemplo que será utilizado no passo de saturação e na ter-

ceira linha é apresentado o exemplo selecionado; neste exemplo [eastbound(east1)].

O segundo passo é a construção da cláusula mais espećıfica que implica o exemplo

selecionado no passo anterior e que esteja de acordo com as restrições da linguagem.

Normalmente é uma cláusula com muitos literais, a qual, como já mencionado, é

chamada de cláusula bottom e o passo é chamado de saturação:

[bottom clause] eastbound(A) :- has_car(A, B), has_car(A, C),

has_car(A, D), has_car(A, E),short(B), short(D), closed(D),

long(C),long(E), open_car(B), open_car(C), open_car(E), shape(B,

rectangle), shape(C, rectangle), shape(D, rectangle), shape(E,

rectangle),wheels(B, 2), wheels(C, 3), wheels(D, 2), wheels(E, 2),

load(B, circle, 1), load(C, hexagon, 1), load(D, triangle, 1),

load(E, rectangle, 3).

[literals] [25]

[saturation time] [0.0200288]}

[literals] indica o número de literais presentes na cláusula bottom e [saturation time]

o tempo gasto para a construção da cláusula bottom.

Para generalizar apenas um único exemplo, utiliza-se o comando sat(V), onde V é

o número do exemplo que se deseja generalizar. Esse comando apenas encontrará

a cláusula bottom para o exemplo escolhido. Caso se deseje encontrar a best clause

gerada por essa cláusula, utiliza-se o comando reduce. Depois da redução, pode-

se ainda utilizar os comandos covers e coversn que mostram, respectivamente, os

exemplos positivos e os exemplos negativos cobertos pela best clause.

O terceiro passo é a busca, implementada por um algoritmo Branch-and-Bound. Nesse

passo encontra-se uma cláusula mais geral que a cláusula bottom. Isso é feito buscando-

se por algum subconjunto de literais da cláusula bottom que tenha melhor score, ou

seja, cubra o maior número de exemplos positivos e o menor número de negativos.

Como mencionado, esse é o passo chamado de redução:

[reduce]

[best label so far] [[1, 0, 2, 1]/0]

eastbound(A).

[5/5]

eastbound(A) :-

has_car(A, B).

[5/5]

eastbound(A) :-

has_car(A, B).

[5/5]

eastbound(A) :-

has_car(A, B).

...

[-------------------------------------]

[found clause]

eastbound(A) :-

has_car(A, B), short(B), closed(B).

[pos-neg] [5]

[clause label] [[5, 0, 4, 5]]

[clauses constructed] [100]

[-------------------------------------]

[clauses constructed] [100]

[search time] [0.0801152]

[best clause] eastbound(A) :-

has_car(A, B), short(B), closed(B).

[pos-neg] [5]

[atoms left] [0]

}

No passo de redução, como já mencionado, busca-se a cláusula com melhor score e a

cada cláusula constrúıda nessa busca é apresentado o número de exemplos positivos

e negativos cobertos pela cláusula. Na execução acima [5/5] indica que a cláusula

cobre 5 exemplos positivos e 5 negativos.

Em [foundclause] é apresentada uma cláusula com um bom score. [pos-neg] apre-

senta a cobertura da cláusula se a função de avaliação é cobertura (coverage). Por

exemplo, se a função de avaliação utilizada é entropia, então será apresentado o nome

da função [entropy] e o valor para a cláusula.

[clause label] é apresentada como [[P,N,L,Val]], onde P e N representam, respec-

tivamente, o número de exemplos positivos e negativos cobertos pela cláusula; L indica

o número de literais da cláusula sendo que o valor de L é menor ou igual ao valor do

parâmetro clauselength; Val indica o score da cláusula.

[clauses constructed] indica o número de nós de busca explorados para encontrar

a cláusula. [best clause] apresenta a melhor cláusula encontrada, entre todas as

[found clause] encontradas na busca, e que será adicionada à teoria final seguida

da sua cobertura ([pos-neg]) e [atoms left] indica o número de exemplos positivos

que não foram cobertos pela cláusula.

O quarto passo é a remoção dos exemplos redundantes. A cláusula com melhor score
é adicionada à teoria e todos os exemplos positivos cobertos por essa cláusula são
removidos. Esse passo é chamado de cover removal e o resultado final será a teoria
encontrada. Também é apresentada a matriz de confusão utilizando o próprio conjunto
de treinamento:

[theory]

[Rule 1] [Pos cover = 5 Neg cover = 0]

eastbound(A) :-

has_car(A, B), short(B), closed(B).

[Training set performance]

Actual

+ -

+ 5 0 5

Pred

- 0 5 5

5 5 10

Accuracy = 1

[Training set summary] [[5, 0, 0, 5]]

[time taken] [0.110158]

[total clauses constructed] [100]

Para a teoria encontrada são apresentadas as regras e, para cada regra, são apresen-

tados os valores de Pos cover e Neg cover que indicam, respectivamente, o número

de exemplos positivos e negativos cobertos pela regra. [Training set performance]

apresenta a matriz de confusão na seguinte forma:

[Training set performance]

Actual

+ -

+ TP FP PP

Pred

- FN TN PN

P N Total

onde TP e FP representam, respectivamente, os verdadeiros positivos e falsos positivos,

e TN e FN os verdadeiros negativos e falsos negativos; P indica o número total de

exemplos positivos (P=TP+FN ) e N o número total de exemplos negativos (N=FP+TN);

PP representa o número de exemplos classificados como positivos (PP=TP+FP) e PN o

número de exemplos classificados como negativos (PN=FN+TN); Total indica o número

total de exemplos (Total=PP+PN); [Training set summary] representa a matriz de

confusão resumida [[TP,FP,FN,TN]]; [time taken], o tempo total para construir

a teoria e [total clauses constructed], o número de cláusulas constrúıdas até

encontrar a teoria final.

Pode-se armazenar a teoria encontrada em um arquivo, alterando o parâmetro set(record, V),

cujo valor padrão é V = false. Se V = true, deve-se fornecer o caminho do arquivo

através do parâmetro :-set(recordfile,V), onde V é o caminho do arquivo.

3 Uso Avançado do Aleph

Aleph permite construir uma teoria por meio do algoritmo básico, descrito na Seção 2

na página 5. Como dito anteriormente, o algoritmo básico é executado por meio do

comando induce/0. Além do comando básico, Aleph possui várias outras formas de

construir uma teoria. A seguir são apresentados os comandos dispońıveis para induzir

uma teoria.

• induce/0 - Constrói uma cláusula por vez. A cada best clause encontrada, todos

os exemplos positivos cobertos por ela são removidos (greedy cover removal);

• induce_cover/0 - Similar ao induce/0, a diferença é que os exemplos positivos

cobertos por cada best clause não são removidos (sem greedy cover removal). A

teoria constrúıda por induce/0 e induce_cover/0 é dependente da ordem em

que os exemplos positivos são apresentados. Uma comparação entre esses dois

comandos é apresentada na Seção 3.1;

• induce_max/0 - Constrói uma cláusula por vez, e assim como o induce_cover/0

não remove os exemplos positivos cobertos por cada best clause. A diferença é

que os passos de saturação e redução são realizados para todos os exemplos, ou

seja, esses passos continuam para todos os exemplos, mesmo que já se tenha

encontrado cláusulas que cobrem todos os exemplos positivos. Devido a essa

caracteŕıstica, a ordem dos exemplos positivos não afeta a teoria final. Para

tornar a busca mais eficiente, memoriza-se a cobertura da best clause encontrada

para cada exemplo e, assim, Aleph constrói um conjunto máximo de cobertura

como solução;

• induce_theory/0 - Não constrói uma cláusula por vez como nos casos anteriores.

Este comando somente é utilizado quando a função de busca é rls, ou seja, para

os métodos de busca aleatórios. Detalhes de como utilizar esses métodos e como

eles funcionam encontram-se na Seção 3.4 na página 42;

• induce_tree/0 - Comando utilizado para a construção de árvores de decisão

relacionais. Detalhes estão descritos na Seção 3.2 na página 29;

• induce_constraints/0 - Comando utilizado para realizar a indução de restri-

ções. Esse comando emula o sistema Claudien (Dehaspe et al., 1996), encon-

trando todas as restrições baseado no conhecimento de fundo. Maiores detalhes

podem ser encontrados na Seção 3.3 na página 35;

• induce_incremental/0 - Constrói a teoria com uma forma de aprendizado in-

cremental, ao contrário dos outros comandos dispońıveis que realizam aprendi-

zado em batch, no qual todos os exemplos e o conhecimento de fundo devem ser

fornecidos antes que o aprendizado comece.

• induce_modes/0 - Faz a extração de modos e tipos automaticamente a partir do

conhecimento do domı́nio e declarações dos determinations fornecidos.

Aleph também possui várias estratégias de busca que podem ser utilizadas, tais como,

ar que implementa uma forma simplificada de regras de associação — Seção 3.3 —

, rls e scs para serem utilizadas pelos métodos de busca aleatória — Seção 3.4.

Este parâmetro é ajustado utilizando set(search,Strategy). Alguns dos métodos

de busca dispońıveis são apresentados ao longo deste trabalho. Detalhes de outras

estratégias podem ser encontradas em (Srinivasan, 2000).

Além da forma como as teorias podem ser constrúıdas e das estratégias de busca,

Aleph permite modificar a função de avaliação utilizada. Este parâmetro é ajus-

tado utilizando set(evalfn,V). V pode ser coverage, compression, posonly, pbayes,

accuracy, laplace, auto_m, mestimate, entropy, gini, sd, wracc ou user (padrão

é coverage). Maiores detalhes sobre as funções de avaliação podem ser encontrados

em (Srinivasan, 2000).

Dessa maneira, por meio dos diversos parâmetros que Aleph tem dispońıvel, além da

caracteŕıstica de permitir que se altere as funções de busca, funções de avaliação e

dos diferentes comandos para construir as teorias, é posśıvel emular alguns sistemas de

PLI, construindo diferentes teorias, tais como, regras de associação relacionais, árvores

de decisão relacionais e indução de restrições, entre outras.

3.1 Comparação entre a Construção de Teorias induce/0 e

induce cover/0

A diferença entre os comandos induce/0 e induce_cover/0 pode ser crucial para o

aprendizado do conceito. Considere o conjunto de exemplos na Tabela 2. Claramente,

uma posśıvel hipótese correspondente a esses exemplos seria:

Se At-1 ≤ 4 então classe = +.

Considere o seguinte arquivo de conhecimento do domı́nio .b:

Ex. At-1 Classe
1 2 +
2 3 +
3 4 +
4 6 -
5 7 -
6 8 -

Tabela 2: Conjunto de exemplos

:-modeh(1,class(+att_id,+class)).

:-modeb(*,peso(+att_id,-peso)).

:-modeb(*,menor_igual(+peso,#peso)).

:-determination(class/2,peso/2).

:-determination(class/2,menor_igual/2).

class(pos).

class(neg).

menor_igual(X,Y):-

var(Y), !,

X = Y.

menor_igual(X,Y):-

number(X), number(Y),

X =< Y.

peso(1,2).

peso(2,3).

peso(3,4).

peso(4,6).

peso(5,7).

peso(6,8).

peso(7,4).

o arquivo de exemplos positivos .f

class(1,pos).

class(2,pos).

class(3,pos).

e o arquivo de exemplos negativos .n.

class(4,neg).

class(5,neg).

class(6,neg).

Executando o comando básico induce/0, o conceito não é aprendido. Isso acontece

porque esse comando remove os exemplos positivos cobertos por cada best clause encon-

trada. Analisando a execução, é posśıvel observar que Aleph constrói, para o primeiro

exemplo positivo, a seguinte cláusula bottom:

class(A,B) :-

peso(A,C), menor_igual(C,2).

O processo de redução dessa cláusula não encontrará uma regra que cubra mais exem-

plos positivos, pois como pode ser observado na Tabela 2, todos os outros têm peso

maior que 2. E assim, a best clause encontrada pelo sistema (que cobre o exemplo e

tem o menor número de literais) é class(1,pos), ou seja, o próprio exemplo. Como

ainda faltam dois outros exemplos a serem cobertos, satura-se o segundo exemplo, cuja

cláusula bottom encontrada é:

class(A,B) :-

peso(A,C), menor_igual(C,3).

Com essa cláusula, o sistema poderia encontrar uma regra que cobrisse também o

primeiro exemplo positivo class(1,pos), que tem peso menor que 3. Porém, esse

exemplo foi removido da base de exemplos positivos, pois o sistema já encontrou uma

best clause para cobri-lo. Novamente, a best clause encontrada pelo sistema cobre

apenas o exemplo atual class(1,pos). O mesmo acontece com o próximo exemplo

class(3,pos), que poderia cobrir os outros 2 exemplos se o sistema não os tivesse

removido da base de exemplos positivos. Como consequência, a teoria encontrada pelo

sistema consiste simplesmente dos exemplos positivos.

[theory]

[Rule 1] [Pos cover = 1 Neg cover = 0]

class(1,pos).

[Rule 2] [Pos cover = 1 Neg cover = 0]

class(2,pos).

[Rule 3] [Pos cover = 1 Neg cover = 0]

class(3,pos).

[Training set performance]

Actual

+ -

+ 3 0 3

Pred

- 0 3 3

3 3 6

Utilizando o comando induce/0, o sistema induziria uma teoria mais geral apenas se

o terceiro exemplo class(3,pos) estivesse em primeiro lugar no arquivo .f. Dessa

forma, o sistema ainda não teria removido os exemplos positivos com peso menor ou

igual a 4. Em contrapartida, executando o comando induce_cover/0 obtém-se uma

teoria mais geral.

[theory]

[Rule 1] [Pos cover = 1 Neg cover = 0]

class(1,pos).

[Rule 2] [Pos cover = 2 Neg cover = 0]

class(A,B) :-

peso(A,C), menor_igual(C,3).

[Rule 3] [Pos cover = 3 Neg cover = 0]

class(A,B) :-

peso(A,C), menor_igual(C,4).

[Training set performance]

Actual

+ -

+ 3 0 3

Pred

- 0 3 3

3 3 6

Apesar de parecer mais vantajosa, a execução do Aleph com o comando induce_cover/0

é mais lenta. Além disso, produz cláusulas redundantes, ou seja, que cobrem os mesmos

exemplos. Essa caracteŕıstica torna necessária a utilização de um programa de separa-

ção para encontrar algum subconjunto dessas cláusulas que minimize essa redundância,

como por exemplo o sistema T-Reduce. Uma cópia desse sistema está dispońıvel em:

http://www.comlab.ox.ac.uk/oucl/research/areas/machlearn/Aleph/misc/treduce.pl

3.2 Indução de Árvores de Decisão Relacionais

O sistema Aleph é capaz de fazer a indução de árvores de decisão relacionais por meio

da emulação das funcionalidades do sistema de PLI Tilde (Blockeel, 1998). O Tilde,

versão relacional do C4.5 Quinlan (1988), é um dos mais famosos algoritmos para

indução de árvores relacionais.

Um exemplo de uma árvore relacional é apresentado na Figura 5. Essa árvore prediz

a ação de manutenção A, que deve ser realizada na máquina M (manutenção(M,A)),

baseada nas partes que a máquina possui (temparte(M,X)), sua condição (usado(X)) e

a facilidade de substituição (insubstituivel(X)). O predicado alvo é manutenção(M,A),

a variável classe é A e o conhecimento de fundo consiste dos predicados temparte(M,X),

usado(X) e insubstituivel(X).

Figura 5: Árvore de decisão relacional para predição da variável classe A do predicado
alvo manutenção(M,A)

Árvores de decisão relacionais possuem uma estrutura muito parecida com as árvores

de decisão proposicionais. Os nós internos contêm testes e os nós folha as predições

para os valores da classe. Caso a variável classe seja discreta ou cont́ınua, a árvore será,

respectivamente, para problemas de classificação ou regressão. A árvore da Figura 5 é

uma árvore de classificação relacional. A maior diferença entre as árvores relacionais

e proposicionais está nos testes que aparecem nos nós internos. No caso relacional os

testes são consultas lógicas (queries) (Lavrač & Džeroski, 2001).

As árvores relacionais são árvores binárias. Cada nó interno tem um ramo esquerdo

(sim), ou seja, uma resposta/substituição que torne essa consulta verdadeira, e um

ramo direito (não) caso contrário. É importante notar que as variáveis podem ser

compartilhadas entre os nós. Uma variável introduzida em um nó pode ser referenciada

em uma sub árvore esquerda daquele nó. Por exemplo, na árvore da Figura 5, a

variável X em insubstituivel(X) refere-se a uma parte X da máquina M, introduzida

no teste temparte(M,X), usado(X) no nó ráız. O teste a ser executado em um nó é a

conjunção dos literais do próprio nó com os literais no caminho da ráız até o nó em

questão. No exemplo da Figura 5, os testes que levam ao nó folha A=enviar volta são:

temparte(M,X), usado(X), insubstituivel(X). Em outras palavras, a máquina precisa

ser enviada de volta para manufatura somente se tem alguma parte dela que seja usada

e é insubstituivel (Džeroski, 2003).

Assim, uma árvore de decisão relacional não pode ser transformada em um programa

lógico que consiste de uma cláusula por folha, ao contrário da árvore de decisão propo-

sicional, na qual a transformação de uma regra de decisão por folha é posśıvel. Árvores

relacionais podem ser facilmente transformadas em listas de decisão de primeira or-

dem, ou seja, conjuntos de cláusulas ordenadas (as cláusulas de programas lógicos são

não-ordenadas). Quando se está aplicando uma lista de decisão a um exemplo, sem-

pre utiliza-se a primeira cláusula que se aplica e é produzida uma resposta. Quando

aplicando um programa lógico, todas as cláusulas aplicáveis são usadas e um conjunto

de respostas pode ser produzido.

Uma lista de decisão é produzida utilizando busca em profundidade depth-first para

visitar os nós da árvore relacional, i.é., descendo primeiro pelo ramo da esquerda.

Cada folha representa uma predição e todas as condições ao longo do ramo esquerdo

levam àquela folha. Uma lista de decisão obtida da árvore na Figura 5 é apresentada

a seguir:

manutenção(M,X)← temparte(M,X),usado(X), insubstituivel(X), !, A=enviar volta.

manutenção(M,X)←temparte(M,X),usado(X),!, A=reparar casa.

manutenção(M,X)← A=sem manutenção.

ordem para assegurar que somente a primeira cláusula aplicável foi utilizada. Para

a primeira cláusula apresentada na lista de decisão, as condições em ambos os nós

internos são válidas, ou seja, todos os ramos que levam à folha enviar volta estão na

esquerda (resposta sim). Na segunda cláusula, como a primeira não foi verdadeira,

para alcançar o nó folha reparar casa, segue-se pelo ramo esquerdo (sim), o que repre-

senta que a condição do nó temparte(M,X) é válida; em seguida esse ramo leva ao nó

insubstituivel(X), no qual o ramo de sáıda que vai levar à folha é o direito (não), o que

representa que a condição do nó insubstituivel(X) não é válida. Assim, para a segunda

cláusula, somente a condição da ráız é válida.

Observe que o algoritmo básico do sistema realiza a classificação de apenas uma classe.

Mesmo que se tenha um problema com mais de uma classe, apenas uma delas será

positiva, e todas as outras serão negativas. A construção de uma teoria baseada em

árvores, porém, classificará todas elas, sendo mais vantajosa para resolver problemas

multiclasses.

Aleph realiza a construção de uma teoria baseada em árvores relacionais utilizando o

comando:

induce_tree.

e baseado em uma abordagem de particionamento recursivo para construir a árvore,

que pode ser de 3 tipos: classificação, regressão ou class probability. O procedimento

básico constrói uma árvore para predizer o argumento de sáıda do predicado alvo.

Dessa forma, a cabeça da cláusula deve ter apenas um único argumento do tipo sáıda.

Além disso, esse argumento não deve estar presente na declaração de tipos, pois os

valores que pode assumir são especificados com o parâmetro :-set(classes, list),

explicado com mais detalhes adiante.

A seguir é apresentado um exemplo no qual, dados nomes de animais, deseja-se predizer

se são mamı́feros ou não. São apresentados alguns exemplos e a declaração do modeh/2.

%Exemplos

class(cachorro,mamifero).

class(gato, mamifero).

class(tartaruga,n_mamifero).

:- modeh(1,class(+animal,-class)).

É importante observar que todos os exemplos são colocados no arquivo .f, e não é

necessário declarar set(evalfn,posonly). No exemplo descrito, tanto os exemplos

da classe mamifero como os da classe n_mamifero são colocados no mesmo arquivo.

Para problemas de classificação, é realizada a poda no conjunto final de cláusulas da

árvore e a técnica utilizada é o método reduced-error proposto por Quinlan (1988).

Para os problemas de regressão, a poda é feita utilizando uma estimativa pessimista

(pessimistic estimate) e para class probability a poda não é utilizada.

No processo de aprendizado com árvores relacionais os caminhos da raiz até as fo-

lhas constituem cláusulas, assim, em todos os casos o resultado é apresentado como

uma seqüência de regras. Para que o Aleph possa induzir a árvore relacional, alguns

parâmetros precisam ser ajustados:

set(tree_type,+V) Este parâmetro ajusta o tipo de árvore que será utilizado no

aprendizado. V é classification, class_probability ou regression;

set(classes,+V) V é uma lista de classes a ser pretida pela árvore. É usada somente

para classificação (classification) ou class probability(class_probability).

Exemplo: set(classes,[mamifero,n_mamifero]).;

set(evalfn,+V) Especifica qual a função de avaliação que será usada na busca. Para

árvores de classificação e class probability esse parâmetro deve ser ajustado para

entropy ou gini. No caso de árvores de regressão, V é ajustado automaticamente

por Aleph para sd (standard deviation);

Além desses parâmetros obrigatórios, há outros parâmetros relevantes, tais como:

set(prune_tree,+V) V é auto ou false (padrão é false). Determina se as regras

constrúıdas utilizando o aprendizado de árvore devem ser podadas utilizando

poda pessimista;

set(mingain,+V) V é um número real (padrão é 0.5). Especifica o ganho mı́nimo

esperado para dividir uma folha durante o processo de construção da árvore;

set(confidence,+V) V é um número real no intervalo de 0.0 a 1.0 (padrão é 0.95).

Determina a confiança para poda de regras no aprendizado com árvores. Se

ajustado para 0.25 é idêntico ao utilizado pelo C4.5;

set(lookahead,+V) V é um número inteiro (padrão é 1). Ajusta um valor para o

operador automático de refinamento, permitindo que sejam adicionados mais que

um literal por vez. Quando induzindo árvores, o parâmetro lookahead é utilizado

para avaliar a qualidade de uma divisão de um nó na busca, considerando não só

a qualidade da cláusula no nó filho, mas também a qualidade dos nós seguintes.

Em outras palavras, ajusta a profundidade analisada na busca antes de realizar

o refinamento;

set(minpos,+V) V é um inteiro positivo (padrão é 1). Para árvores, esse parâmetro

especifica o número mı́nimo de exemplos que deve haver em uma folha para que

ela seja dividida.

Além desses parâmetros que são ajustados pelo usuário, o parâmetro refine é ajus-

tado automaticamente pelo sistema para auto. Além disso, o Aleph versão 5, com

data de modificação do dia 4 de junho de 2006, requer a utilização do parâmetro

set(dependent,+V), onde V é o inteiro positivo que representa o argumento da va-

riável dependente nos exemplos, ou seja, V é o número do argumento do predicado

alvo que representa a classe. Por exemplo, se o argumento que representa a classe é

o primeiro, V = 1. Note que a especificação desse parâmetro fornece mais liberdade

para a elaboração do predicado alvo, pois pode-se indicar qualquer argumento para

representar a classe, não necessariamente um argumento de sáıda.

A seguir é apresentado um exemplo de como utilizar o Aleph para indução de árvo-

res de decisão relacionais. Utilizando o arquivo animals.b, apresentado a seguir, no

qual estão definidos os parâmetros, as declarações de modos, determinations, tipos e

algumas declarações de conhecimento de fundo:

% Especificando o tipo da árvore

:- set(tree_type,classification).

:- set(classes,[mammal,nmammal]).

:- set(minpos,2).

:- set(prune_tree,auto).

:- set(confidence,0.25).

:- set(evalfn,entropy).

:- set(lookahead,2).

:- modeh(1,class(+animal,-class)).

:- modeb(1,has_gills(+animal)).

:- modeb(1,has_covering(+animal,#covering)).

:- modeb(1,has_legs(+animal,#nat)).

:- modeb(1,homeothermic(+animal)).

:- modeb(1,has_eggs(+animal)).

:- modeb(1,not(has_gills(+animal))).

:- modeb(1,nhas_gills(+animal)).

:- modeb(*,habitat(+animal,#habitat)).

:- modeb(1,has_milk(+animal)).

:-determination(class/2,has_gills/1).

:-determination(class/2,has_covering/2).

:-determination(class/2,has_legs/2).

:-determination(class/2,momeotermic/1).

:-determination(class/2,has_egss/1).

:-determination(class/2,nhas_gills/1).

:-determination(class/2,habitat/2).

:-determination(class/2,has_milk/1).

% Declaraç~ao dos Tipos

animal(dog). animal(dolphin). animal(platypus). animal(bat).

animal(trout). animal(herring). animal(shark). animal(eel).

animal(lizard). animal(crocodile). animal(t_rex). animal(turtle).

animal(snake). animal(eagle). animal(ostrich). animal(penguin).

animal(cat). animal(dragon). animal(girl). animal(boy).

class(mammal). class(fish). class(reptile). class(bird).

covering(hair). covering(none). covering(scales). covering(feathers).

habitat(land). habitat(water). habitat(air). habitat(caves).

%Conhecimento de Fundo

has_covering(dog,hair). has_covering(dolphin,none).

has_covering(platypus,hair). has_covering(bat,hair).

has_legs(dog,4). has_legs(dolphin,0). has_legs(platypus,2).

has_milk(dog). has_milk(cat). has_milk(dolphin).

homeothermic(dog). homeothermic(cat). homeothermic(dolphin).

homeothermic(platypus).

habitat(dog,land). habitat(dolphin,water).

habitat(platypus,water). habitat(bat,air).

has_eggs(platypus). has_eggs(trout). has_eggs(herring).

has_eggs(shark).

has_gills(trout). has_gills(herring). has_gills(shark).

has_gills(eel).

nhas_gills(X) :- animal(X), not(has_gills(X)).

Note que todos os tipos foram especificados no arquivo .b. No aprendizado de árvores

relacionais, é necessário que se especifique todos os tipos declarados nos argumentos

dos predicados. Diferentemente do aprendizado básico do sistema, o aprendizado de

árvores ignora predicados que possuem algum de seus tipos (que podem ser de entrada,

sáıda ou constantes) não especificados.

Os exemplos no arquivo animal.f, são apresentado a seguir.

class(bat,mammal).

class(dog,mammal).

class(platypus,mammal).

class(eagle,nmammal).

class(eagle,nmammal).

class(ostrich,nmammal).

class(shark,nmammal).

class(crocodile,nmammal).

class(shark,nmammal).

O resultado final da execução do Aleph é:

[theory]

[Rule 1] [Pos cover = 6 Neg cover = 0]

class(A,B) :-

has_covering(A,hair), B=mammal.

[Rule 2] [Pos cover = 14 Neg cover = 0]

class(A,B) :-

not has_covering(A,hair), B=nmammal.

[Training set performance]

Actual

+ -

+ 20 0 20

Pred

- 0 0 0

20 0 20

Accuracy = 1.0

[Training set summary] [[20,0,0,0]]

É importante saber que todos os experimentos realizados para a construção de árvores

de decisão relacionais utilizando Yap-5.1.1 e Aleph versão 5, com data de modificação

de 6 de novembro de 2005, não geraram teoria.

3.3 Regras de Associação Relacional

Regras de associação constituem uma das técnicas mais utilizadas em mineração de

dados quando não se conhece o predicado alvo do problema. As regras de associação

caracterizam o quanto a presença de um conjunto de itens nos registros (exemplos) de

uma base de dados implica a presença de algum outro conjunto distinto de itens nos

mesmos registros (Dehaspe & Toivonen, 2001). Um exemplo t́ıpico de utilização de

regras de associação proposicionais é apresentado na Figura 6.

Figura 6: Exemplo de uma regra de associação

O formato de uma regra de associação é:

LHS =⇒ RHS [suporte, confiança]

onde LHS (Left Hand Side) representa o lado esquerdo da regra e RHS (Right Hand

Side) o lado direito, sendo que LHS e RHS são definidos por conjuntos disjuntos de

itens. Existem duas medidas que estão sempre associadas a uma regra de associação:

o suporte e a confiança. O suporte quantifica a incidência da regra no conjunto de

dados e representa a freqüência com que LHS e RHS ocorrem juntos no conjunto de

dados. A confiança indica a freqüência com que LHS e RHS ocorrem juntos em relação

ao número total de registros nos quais LHS ocorre, ou seja, representa a probabilidade

de ocorrência de RHS dado que LHS ocorre.

Nos casos em que a base de dados utilizada for relacional, regras de associação relaci-

onal poderão ser descobertas. Nesse caso, consultas Datalog podem ser vistas como a

versão relacional dos itemsets. Um exemplo de consulta é:

? - pessoa(X), pais(X,Y), tem_animal_estimaç~ao(Y,Z).

Considere o itemset {pessoa,pais,filho,animal estimação} para regras de associação

proposicionais. Uma interpretação desse padrão é que pessoa, pais, filho e animal estimação

ocorrem juntos. Este é o mesmo significado da consulta anterior, entretanto, as variá-

veis X, Y e Z adicionam informação extra: pessoa e pais são os mesmos, pais e filhos

pertencem à mesma famı́lia e os animais de estimação pertencem aos filhos. Isso ilustra

o fato de que as consultas são mais expressivas que os itemsets.

A tarefa de descobrir padrões relacionais que estejam associados entre si foi endereçada

ao sistema relacional WARMR (Dehaspe & De Raedt, 1997). O WARMR não gera

itemsets, apenas padrões freqüentes. As regras geradas a partir de consultas freqüentes

são chamadas de consultas estendidas. O WARMR recebe como entrada uma base

de dados, uma freqüência mı́nima e um bias de linguagem L. O bias de linguagem

especifica um átomo chave e modos de entrada e sáıda para os predicados ou relações.

Para a descoberta de padrões freqüentes é necessária a noção de freqüência. Conside-

rando as consultas como padrões e considerando que as consultas podem ter variáveis,

não é muito óbvia qual a freqüênia de uma consulta. Isso é resolvido especificando-se

um parâmetro adicional para a tarefa de descoberta de padrões, chamado de chave. A

chave é um átomo que deve estar presente em todas as consultas durante o processo

de descoberta, pois ele determina o que contar. Se pessoa(X) é a chave, contam-se

pessoas; se pais(X,Y) é a chave, contam-se pares(pais,filhos). Assim, utilizando o pa-

râmetro chave e fazendo essa contagem encontra-se o suporte e a confiança das regras

de associação relacional. Esse parâmetro é importante uma vez que se torna posśıvel

mudar o foco da aplicação sem que as entradas e, em particular, a base de dados, sejam

modificadas (Džeroski, 2003).

Para se obter regras de associação relacionais utilizando o Aleph, pode-se utilizar

uma das duas abordagens existentes. Uma das abordagens é utilizar a busca ar

(set(search,ar)), que é uma emulação do WARMR. Entretando, vale notar que

o Aleph não faz o mesmo que o WARMR. O que o Aleph faz é rotular como good

clause todas as regras encontradas, durante o passo de redução e que cobrem um nú-

mero maior ou igual a uma certa fração dos exemplos positivos. Essa fração é definida

pelo parâmetro set(pos_fraction,V) que deve ter seu valor ajustado entre 0.0 e 1.0.

A seguir é apresentado um exemplo de como utilizar o Aleph para emular o WARMR.

No arquivo customer.b, mostrado a seguir, é definido o bias de linguagem, o conheci-

mento de fundo e os tipos, entre outros. A definição do modeh/2 pode ser vista como

a definição do parâmetro chave utilizada pelo WARMR, e todos os argumentos do

predicado alvo devem ser do tipo de entrada.

:- set(search,ar).

:- set(pos_fraction,0.5).

:- modeh(*,customer(+person)).

:- modeb(1,parent(+person,-person)).

:- modeb(1,parent(-person,+person)).

:- modeb(*,buys(+person,#product)).

:- modeb(*,nbuys(+person,#product)).

:- modeb(1,not(parent(+person,+person))).

:- determination(customer/1,person/1).

:- determination(customer/1,parent/2).

:- determination(customer/1,buys/2).

:- determination(customer/1,(not)/1).

:- determination(customer/1,nbuys/2).

nbuys(A,B):- person(A), product(B),

not(buys(A,B)).

person(allen). person(bill). person(carol).

person(diana). person(zoe).

product(wine). product(cola). product(pizza).

parent(allen,bill). parent(allen,carol).

parent(bill,zoe). parent(carol,diana).

buys(allen,wine). buys(bill,cola).

buys(bill,pizza). buys(diana,pizza).

No caso dessa abordagem, não são utilizados exemplos negativos, apenas exemplos

positivos. Um exemplo desse arquivo é apresentado a seguir no arquivo customer.f.

customer(allen).

customer(bill).

customer(carol).

customer(diana).

Os arquivos são lidos utilizando read_all(costumer) e em seguida a indução pode

ser ativada utilizando o comando induce/0. Ao final são apresentadas as cláusulas

rotuladas como [good clause], a teoria e a matriz de confusão. Perceba que todas

as cláusulas rotuladas como [good clause] cobrem um número maior ou igual que o

valor atribúıdo ao parâmetro pos_fraction multiplicado pela quantidade de exemplos

positivos, ou seja, 2 ou mais exemplos.

[good clauses]

customer(A).

[pos-neg] [4]

customer(A) :-

parent(A,B).

[pos-neg] [3]

customer(A) :-

parent(B,A).

[pos-neg] [3]

customer(A) :-

nbuys(A,wine).

[pos-neg] [3]

customer(A) :-

nbuys(A,pizza).

[pos-neg] [2]

customer(A) :-

parent(A,B), buys(B,pizza).

[pos-neg] [2]

customer(A) :-

parent(A,B), nbuys(B,wine).

[pos-neg] [3]

...

[atoms left] [0]

[positive examples left] [0]

[estimated time to finish (secs)] [0.0]

[theory]

[Rule 1] [Pos cover = 4 Neg cover = 0]

customer(A).

[Training set performance]

Actual

+ -

+ 4 0 4

Pred

- 0 0 0

4 0 4

Accuracy = 1.0

[Training set summary] [[4,0,0,0]]

[time taken] [0.024]

[total clauses constructed] [63]

É importante observar que, quando o objetivo é encontrar regras de associação relacio-

nais, o resultado final ([theory]) não é o importante, mas sim, todas as associações que

o Aleph conseguiu encontrar ao longo da busca, ou seja, as cláusulas [good clause]

que contêm associações verdadeiras de acordo com os exemplos positivos.

A outra abordagem que pode ser utilizada com o Aleph, consiste em encontrar as as-

sociações não verdadeiras contidas na base de dados. Para isso, utiliza-se o comando

induce_constraints/0. Esse comando emula o sistema Claudien (Dehaspe et al.,

1996), encontrando todas as restrições no conhecimento de fundo. As restrições en-

contradas são da forma false:-body. Uma vez que body inclui a negação de literais,

é posśıvel se obter regras com diferentes śımbolos predicados na cabeça.

O exemplo a seguir demonstra o aprendizado de restrições. Note que, para que o

Aleph encontre as restrições, o usuário deve fazer declarações de modo para false

para a cabeça, e not para os predicados do corpo que definem o conceito. O predicado

do corpo que determina o conceito deve ter apenas um argumento de sáıda. Além

disso, como as restrições são baseadas no conhecimento de fundo, exemplos positivos

e negativos não são utilizados. Os valores apresentados por [pos/neg] indicam que o

Aleph considera os dados do conhecimento do domı́nio como exemplos negativos (já

que se deseja negar aquele conhecimento), e não há exemplos positivos.

:- modeh(1,false). %declaraç~ao de modo para false

:- modeb(*,humano(-pessoa)). %conceito humano, com apenas uma argumento de saı́da

:- modeb(1, homem(+pessoa)). %humano é definido pelos predicados homem e mulher

:- modeb(1, mulher(+pessoa)).

:- modeb(1, not(homem(+pessoa))). %declaraç~oes de modo para not

:- modeb(1, not(mulher(+pessoa))).

:- determination(false/0,human/1).

:- determination(false/0,male/1).

:- determination(false/0,female/1).

:- determination(false/0,(not)/1).

male(’Fred’). %conhecimento de fundo

female(’Wilma’).

human(’Fred’).

human(’Wilma’).

As restrições que Aleph encontra são:

[constraints]

false :-

human(A), male(A), female(A).

[pos-neg] [0]

false :-

human(A), female(A), male(A).

[pos-neg] [0]

false :-

human(A), not male(A), not female(A).

[pos-neg] [0]

false :-

human(A), not female(A), not male(A).

[pos-neg] [0]

[total clauses constructed] [27]

A primeira e a segunda regras são equivalentes e indicam que não existem humanos

que sejam homem e mulher ao mesmo tempo. A terceira e a quarta regras são também

equivalentes e indicam que não existem humanos que não sejam nem homens nem

mulheres. Encontradas as restrições, o usuário pode, então, remover as restrições

redundantes e transformar restrições (transladando para a cabeça os predicados com

not). Nesse caso, obtém-se as seguintes regras equivalentes às anteriores:

%n~ao existe humano que seja homem e mulher ao mesmo tempo

false:- humano(A),homem(A),mulher(A).

%um homem ou uma mulher é um humano

homem(A);mulher(A):- humano(A).

Abaixo é apresentado outro exemplo, em que se deseja encontrar restrições para o

conceito ruivo.

:- modeh(1,false).

:- modeb(1,ruivo(-person)).

:- modeb(1,cor_cabelo(+person, #cor)).

:- modeb(1,not(cor_cabelo(+person, #cor))).

:- determination(false/0,ruivo/1).

:- determination(false/0,cor_cabelo/2).

:- determination(false/0,(not)/1).

person(fred).

person(wilma).

cor(vermelho).

cor(amarelo).

cor(castanho).

ruivo(fred).

ruivo(wilma).

cor_cabelo(fred, vermelho).

cor_cabelo(wilma, vermelho).

Algumas das restrições encontradas são:

[constraints]

false :-

ruivo(A), cor_cabelo(A,amarelo).%ruivos n~ao têm cabelo amarelo

[pos-neg] [0]

false :-

ruivo(A), cor_cabelo(A,castanho).%ruivos n~ao têm cabelo castanho

[pos-neg] [0]

false :-

ruivo(A), not cor_cabelo(A,vermelho).%ruivos n~ao devem n~ao ter cabelo vermelho

O resultado das duas abordagens são associações encontradas na bases de dados, ver-

dadeiras, ou falsas no caso da segunda abordagem apresentada. Os resultados dessas

duas abordagens podem ser combinados para encontrar associações interessantes na

base de dados.

3.4 Métodos de Busca Estocástica

A forma mais simples de busca aleatória é: dada uma amostra de N elementos (cláusu-

las ou teorias) do espaço de busca, pontue (score) esses elementos e retorne o melhor.

Um estudo sobre busca aleatória em PLI pode ser encontrado em (Srinivasan, 1999).

Para realizar a seleção estocástica de cláusulas, deve-se ajustar o seguinte parâmetro:

set(search,scs). O número N é ajustado pelo valor associado a scs_sample/2 ou

calculado automaticamente por:

set(scs_prob,P)

set(scs_percentile,K)

Assim, o tamanho da amostra é calculado por:

log(1−P )

log(1− K
100)

Essa fórmula calcula o número de cláusulas que devem ser amostradas, com probabili-

dade de no mı́nimo P , em pelo menos uma cláusula nos K-porcento de cláusulas. Isso

é especificado ajustando o set/2 de scs_type para blind ou informed. Se blind,

então as cláusulas são selecionadas uniformemente no espaço de todas as posśıveis

hipóteses. Se informed, então são dirigidas para uma distribuição espećıfica para o

tamanho das cláusulas. O tamanho das cláusulas, i.é, o número de sentenças no corpo

da cláusula, pode ser pré-especificado pelo parâmetro clauselength_distribuition

ou, quando esse tamanho não é declarado, então ele é obtido automaticamente por um

esquema similar ao Monte-Carlo.

A cláusula inicial aleatoriamente selecionada é freqüentemente constrúıda como segue:

1. Um exemplo é selecionado. O exemplo pode ser selecionado automaticamente

pelo sistema ou pelo usuário usando o comando sat;

2. A cláusula bottom é constrúıda para o exemplo selecionado;

3. Uma cláusula legal é aleatoriamente dirigida para a cláusula bottom.

O sistema pode ser ajustado para não construir a cláusula bottom ajustando-se o

parâmetro construct_bottom para false; nesse caso a cláusula selecionada pode ser

o resultado de uma distribuição sobre o tamanho das cláusulas definido pelo parâmetro

// clauselength_distribuition.

O método estocástico de busca local (randomised local search) usa os algoritmos

GSAT (Éric Grégoire & Mazure, 1998), WSAT (Selman et al., 1994), RRR (Ze-

lezny et al., 2002) ou Annealing (Kirkpatrick, Gelatt, & Vecchi, Kirkpatrick et al.)

para busca em PLI. Para utilizar esse método, a função de busca é ajustada para:

set(search,rls) e o tipo de busca aleatória especificado pelo parâmetro rls_type,

que deve ser ajustado para gsat, wsat, rrr ou anneal respectivamente. Quando a

busca é rls utiliza-se o comando induce_theory/0.

Os algoritmos GSAT, WSAT e Annealing utilizam múltiplos reińıcios, e cada um desses

reińıcios serve como ponto inicial para movimentos locais no espaço de busca. O limite

do número de reińıcios é especificado pelo parâmetro tries, ajustado com um valor

inteiro positivo, e o número de movimentos por moves, ajustado com valor ≥ 0. A

implementação do algoritmo WSAT requer ainda um random-walk probability, o qual

é especificado pelo parâmetro walk com um valor entre 0 e 1. Se walk for igual a 0 o

WSAT será equivalente ao algoritmo GSAT.

O algoritmo de busca Randomised Rapid Restarts (RRR) é descrito em (Zelezny et al.,

2002). RRR pode utilizar dois diferentes critérios de parada:

1. O número mı́nimo de exemplos positivos que devem ser encontrados por uma

cláusula (minpos) e se a utilidade aceitável é alcançada (minscore);

2. A precisão mı́nima é alcançada (minacc). Nesse caso a função de avaliação deve

ser precisão (set(evalfn,accuracy)).

O algoritmo annealing é restringido a usar uma temperatura fixa; a temperatura é

ajustada pelo parâmetro temperature com valor real > 0.

3.5 Aprendizado Abdutivo

O aprendizado abdutivo pode ser utilizado para completar definições do conhecimento

de fundo, assumindo que o conhecimento de fundo está correto, porém incompleto.

Esse aprendizado está incorporado no comando induce/0, bastando apenas alterar o

parâmetro abduce para true (padrão é false) para que seja ativado. Para declarar

candidatos à abdução, utiliza-se o comando abducible(N/A), onde N é o nome do

predicado e A é sua aridade.

Considere o seguinte conhecimento de fundo (arquivo .b):

:-modeh(1, grandfather(+person, -person)).

:-modeb(*, father(+person, -person)).

:-modeb(*, mother(+person, -person)).

:-modeb(*, parent(+person, -person)).

:-determination(grandfather/2, father/2).

:-determination(grandfather/2, mother/2).

:-determination(grandfather/2, parent/2).

:-set(abduce, true).

:-abducible(parent/2).

person(jane).

person(robert).

person(peter).

person(fred).

father(fred, jane).

mother(jane, robert).

mother(jane, peter).

grandfather(X,Y):-father(X,Z), parent(Z,Y).

parent(X,Y):-father(X,Y).

e os seguintes exemplos positivos (arquivo .f):

grandfather(fred,robert).

grandfather(fred,peter).

e negativos (arquivo .n):

grandfather(robert,peter).

grandfather(jane,peter).

Primeiramente, o sistema tenta encontrar outra cláusula para grandfather/2, resul-

tando em:

grandfather(A,B) :-

father(A,C), mother(C,B).

Então, o sistema encontra cláusulas que definam os predicados declarados como abducible.

Essas cláusulas têm mais utilidade que a cláusula encontrada acima, pois elas comple-

tam o conhecimento do domı́nio. A cláusula encontrada para o predicado parent/2

foi:

parent(A,B) :-

mother(A,B).

Para melhorar a eficiência, pode-se limitar o número máximo de átomos ground em

qualquer abdução através do comando :-set(max_abducibles, +V), onde V é um

inteiro (padrão é 2).

3.6 Aprendizado de Features

O aprendizado de features pode ser executado através do comando induce_features/0.

Seu procedimento é muito parecido com o induce_cover/0, pois também não remove

os exemplos positivos a cada best clause encontrada. Porém, cada good clause en-

contrada é adicionada como uma nova feature. Esse processo continua até que as

regras encontradas cubram todos os exemplos positivos ou até que a quantidade de

features encontradas ultrapasse um certo limite. Esse limite é definido pelo parâmetro

set(max_features,V), onde V é um inteiro positivo (padrão é infinito).

Cada feature encontrada contém duas informações: um número identificador da fea-

ture e uma cláusula (Head:-Body). Uma feature é verdadeira para um exemplo, se e

somente se, o exemplo unifica com a cabeça e o corpo é verdadeiro. Para problemas de

classificação, o usuário deve especificar a variável dependente. Isso é feito através do

parâmetro set(dependent,V), onde V é um inteiro positivo que indica o argumento

da variável dependente nos exemplos. As features têm a seguinte forma:

feature(ID,(Head:-Body)).

A rotulação de cláusulas como good clause depende de vários parâmetros do sistema.

Abaixo são apresentados alguns parâmetros relevantes nesse processo:

:- set(clauselength,V): V é um inteiro positivo (padrão é 4). Limita o número

máximo de literais de uma cláusula aceitável;

:- set(minacc,V): V é um real entre 0.0 e 1.0 (padrão é 0.0). Limita a mı́nima precisão

de uma cláusula aceitável. A precisão de uma cláusula tem o mesmo significado que

precisão, ou seja, p/(p+n), onde p é o número de exemplos positivos cobertos pela

cláusula (os verdadeiros positivos) e n é o número de exemplos negativos cobertos pela

cláusula (os falsos positivos);

:- set(minscore,V): V é um real (padrão é infinito). Limita a mı́nima utilidade de

uma cláusula aceitável. A utilidade da cláusula é definida pela função de avaliação

adotada. Detalhes sobre como definir a função de avaliação encontram-se na Seção 3

na página 24;

:- set(minpos,V): V é um inteiro positivo (padrão é 1). Limita o número mı́nimo de

exemplos positivos cobertos por uma cláusula. Esse parâmetro é útil para se evitar

que o Aleph adicione à teoria cláusulas com um único literal;

:- set(noise,V): V é um inteiro positivo, incluindo zero (padrão é 0). Limita o

número máximo de exemplos negativos que podem ser cobertos por uma cláusula

aceitável;

:- set(nodes,V): V é um inteiro positivo (padrão é 5000). Limita o número máximo

de nós a serem explorados durante a busca por uma cláusula aceitável.

:- set(explore,V): V pode ser true ou false (padrão é false). Se verdadeiro, força

o sistema a continuar a busca até que todos os elementos restantes da busca sejam

piores que o melhor elemento atual. Quando é falso, a busca pára quando é certo que

todos os elementos restantes da busca não são melhores que o melhor elemento atual.

(Observe que com false, o critério é mais suave).

É importante observar que as cláusulas que não respeitam os limites definidos por um

parâmetro não são adicionadas à teoria. No caso do aprendizado de features, essas

cláusulas não são adicionadas como uma nova feature.

3.7 Definições do Usuário

O Aleph permite que o usuário defina restrições de integridade, de refinamento, de

poda e de custo, descritas a seguir.

3.7.1 Restrições de Integridade

O sistema Aleph permite que o usuário defina restrições de integridade. Dessa forma,

todas as hipóteses que violarem as restrições serão consideradas inconsistentes e serão

recusadas. Note que essas restrições funcionam como os exemplos negativos, sendo

úteis para encontrar hipóteses interessantes quando se tem apenas exemplos positivos.

A declaração das restrições se dá no arquivo .b e tem a seguinte forma:

false:-

Body.

onde Body, o corpo da restrição, é o conjunto de literais que especificam as con-

dições que não devem ser violadas. Nesse corpo, geralmente se utiliza o comando

hypothesis/3 para obter a cláusula em construção no momento. Sua estrutura é a

seguinte:

hypothesis(-Head,-Body,-Label)

onde Head e Body são, respectivamente, a cabeça e o corpo da cláusula sendo cons-

trúıda, e Label é a lista [P,N,L], sendo P a quantidade de exemplos positivos cobertos

pela cláusula em construção, N a quantidade de exemplos negativos cobertos pela cláu-

sula em construção e L a quantidade de literais. Note que uma restrição de integridade

não impede que o refinamento de uma cláusula que viola uma ou mais restrições seja

inaceitável.

3.7.2 Refinamento

O sistema Aleph permite que o usuário defina como se dará o refinamento durante a

etapa de busca através do comando refine/2. O comando é declarado no arquivo .b

e tem a seguinte forma:

refine(Clause1,Clause2):-

Body.

onde a cláusula Clause1 será refinada para a cláusula Clause2, caso o corpo Body seja

verdadeiro, ou seja, Clause2 será a cláusula mais espećıfica. O conjunto de refinamen-

tos para qualquer cláusula é obtido através de backtracking.

3.7.3 Poda

Quando se sabe que certas cláusulas (e seus refinamentos) não devem aparecer no

conjunto de hipóteses, o usuário pode definir condições de poda para elas. Essa é uma

maneira de fornecer informações sobre o espaço de hipóteses, aumentando a eficiência

do sistema. As definições devem ser declaradas no arquivo .b e são da seguinte forma:

prune((ClauseHead:-ClauseBody)):-

Body.

significando que, se a condição Body for satisfeita, a cláusula ClauseHead:-ClauseBody

será podada.

3.7.4 Custo

Aleph permite que o usuário especifique custo para as cláusulas. A especificação dos

custos deve ser declarada no arquivo .b, e tem a seguinte forma:

cost(Clause,ClauseLabel,Cost):-

Body.

onde ClauseLabel é a lista [P,N,L], onde P é o número de exemplos positivos cobertos

pela cláusula, N é o número de exemplos negativos cobertos pela cláusula e L é o

número de literais da cláusula. Quando o usuário especifica custo, Aleph deixa a cargo

do usuário a especificação de estratégias de poda.

4 Conclusão

Os sistemas de Programação Lógica Indutiva vêm sendo cada vez mais utilizados

como método de extração de conhecimento para Mineração de Dados Relacional. Os

sistemas mais utilizados são os que induzem regras de classificação relacionais. No

entanto, atualmente já existem sistemas se PLI para indução de árvores de decisão

relacionais, regras de associação relacionais e indução de restrições, entre outros.

Neste trabalho foram estudadas caracteŕısticas do sistema de PLI Aleph mostrando

como ele funciona e como utilizá-lo. Aleph é um sistema de PLI que foi desenvolvido

com o intuito de ser um protótipo para explorar idéias. Aleph possui caracteŕısticas

importantes para o desenvolvimento de pesquisas em PLI, pois permite ao usuário

modificar a ordem de geração das regras, funções de busca e avaliação de acordo as

necessidades da aplicação que o usuário deseja desenvolver. Além da indução de re-

gras de classificação relacionais, Aleph permite ainda a indução de árvores de decisão

relacionais, disponibiliza métodos de busca estocásticos, induz restrições e uma forma

simplificada de regras de associação relacionais. Essas abordagens podem ser utiliza-

das por meio do ajuste de parâmetros espećıficos para cada uma delas. Neste trabalho

foram apresentadas cada uma dessas abordagens exemplificando como e quais parâme-

tros ajustar e apresentando o resultado das execuções. Aleph disponibiliza ainda outros

recursos que não foram abordados neste trabalho mas que podem ser encontrados no

manual do Aleph (Srinivasan, 2000).

